
Adaptacyjnego Przetwarzania Sygnałów
Adaptacyjne usuwanie przydźwięku sieci z systemu nagłośnieniowego

Przydźwięk sieciowy jest jednym z problemów występujących w rozległych instalacjach audio. Walka z nim jest bardzo trudna, a niekiedy niemożliwa, szczególnie w przenośnych systemach, np. podczas koncertów lub wystąpień w pobliżu silnych zakłóceń elektromagnetycznych, gdzie nawet zastosowanie dobrych kabli ekranowanych niewiele poprawia sytuację. Można w tym celu stosować filtry wycinające 50Hz, jednakże wraz z zakłóceniami wycinane są niskie częstotliwości sygnału, co nie do pomyślenia jest w przypadku muzyki. Rozwiązaniem tego problemu są systemy adaptacyjne, szczególnie że mamy tutaj w miarę dobrze zdefiniowany sygnał zakłócający – wystarczy pobrać sygnał z sieci energetycznej i zastosować jako sygnał odniesienia. Budowę systemu adaptacyjnego można oprzeć na jednym z wielu algorytmów. Wybraliśmy LMS ze względu na jego prostotę i małe wymagania obliczeniowe i wystarczające w tym zastosowaniu parametry. Drugim powodem naszego wyboru jest to, że posiadany przez nas procesor sygnałowy ma stosunkowo małą wydajność oraz małą pojemność pamięci (500 słów).

Budowę układu oparliśmy o płytkę z procesorem DSP56004, pochodzącą z demontażu. Oprócz DSP znajdują się na niej dwa 18-bitowe przetworniki stereo C/A, jeden przetwornik 18-bitowy A/C oraz pamięć SRAM 32kB która nie jest używana w tej aplikacji, ale ma ona ze strony procesora wsparcie sprzętowe ułatwiające implementację linii opóźniających. Aby zmusić moduł do pracy, dołączono do niego zewnętrzną pamięć EEPROM, układ RESET’u i wyboru trybu ładowania programu oraz zasilacz.

Jako sygnał odniesienia użyto napięcia panującego na wyjściu transformatora zmniejszonego przez dzielnik rezystancyjny . W przypadku pojawienia się w sygnale składowych harmonicznych 100Hz istnieje możliwość wybrania jako sygnału odniesienia, wyprostowanego dwupołówkowo napięcia sieci, które zawiera składowe zarówno 50Hz, 100Hz jak i wyższe.

Dla celów testowych wyposażono układ w możliwość celowego wprowadzenia zakłóceń do sygnału wejściowego

Schemat blokowy urządzenia

Wygląd układu po zmontowaniu i uruchomieniu

BADANIA:

We wszystkich przypadkach przyjęto (=0.005
Widmo sygnału na wejściu filtru adaptacyjnego bez sygnału użytecznego (sam sygnał zakłóceń) 

Widmo sygnału na wyjściu filtru, bez sygnału użytecznego, rząd filtru=120

Widmo sygnału użytecznego (muzyka) wraz z sygnałem zakłócającym na wejściu układu (FFT z nieskończonym uśrednianiem)

Widmo sygnału użytecznego (muzyka) wraz z sygnałem zakłócającym na wyjściu układu (FFT z nieskończonym uśrednianiem), rząd filtru=120

Widmo sygnału na wyjściu filtru, bez sygnału użytecznego, rząd filtru = 200

Widmo sygnału na wyjściu filtru, bez sygnału użytecznego, rząd filtru = 20

 Widmo sygnału użytecznego (muzyka) wraz z sygnałem zakłócającym na wyjściu układu (FFT z nieskończonym uśrednianiem), rząd filtru = 200

Wnioski:

Układ spełnia swoją rolę , w optymalnych warunkach, (przy braku sygnału) i wykorzystaniu maksymalnych możliwości procesora (rząd filtru=200), tłumienie sygnału zakłóceń osiągnęło wartość ponad 50dB.

W przypadku sygnału rzeczywistego, jest ona na tyle dobra, że podczas testów odsłuchowych nie stwierdzono występowania sygnału zakłócającego.

Układ bez żadnych przeróbek można wykorzystać do usuwania dowolnych zakłóceń, o ile jesteśmy w stanie podłączyć sygnał w którym dominuje składowa, którą chcemy wyeliminować.

Ponieważ algorytm wykorzystuje zaledwie 1/3 pojemności, pamięci możliwe jest zaimplementowanie dodatkowo układu przesuwającego pasmo sygnału o kilka Hz, a przez to wyeliminowanie możliwości wzbudzenia się aparatury – następny problem występujący w systemach nagłośnieniowych.

Kod źródłowy:

L equ 120

  ;deklaracja rzedu filtru

; alfa

alfa equ 0.005

  ;deklaracja kroku adaptacji

  org p:$0 



  jmp start1                           ;skok do wlaściwego programu

  org p:$64

start1:

    movep #$260035,x:pctl    ; uaktywnienie PLL i ustawienie zegara DSP na 50MHz

    movep #$000201,x:rcs     ; uaktywnienie układu SAI odbiorczego i ustawienie go w tryb SLAVE

    movep #$000203,x:tcs     ; uaktywnienie układu SAI nadawczego i ustawienie go w tryb MASTER

    movep #$000c00,x:ipr     ; ustawienie przerwań

        ; bufor danych

        move #0,r0

    ;początek bufora danych od zera w pamięci X

        move #L-1,m0
    ;arytmetyka MOD L dla wskaźnika r0


        ; bufor współczynnikow

        move #0,r4 

    ;początek  bufora współczynników od zera w pamięci Y

        move #L-1,m4
    ;arytmetyka MOD L dla wskaźnika r0


                .loop #L
    ; pętla ustawiająca warunki początkowe algorytmu na 0




                        move #0,a

                        move a,y:(r4)+

                .endl

wait_loop:                               ;pętla oczekująca na dane

    jset #rldf,x:rcs,receive_l     ;jeśli dane w buforze kanału lewego, to skocz do programu obsługi

    jset #rrdf,x:rcs,receive_r     ; jeśli dane w buforze kanału prawego, to skocz do programu obsługi

    jmp wait_loop

receive_r:

   bCLR #gdd1,x:gpior               ;sygnalizacja działania programu

    move x:rx0,a              
         ;pobierz próbkę sygnału odniesienia

    move a,x0   
       
         ;zachowaj próbkę dla dalszego przetwarzania

    move a,x:tx0              
         ;wyślij próbkę na wyjście dla celów testowych

    bSET #gdd1,x:gpior               ; sygnalizacja działania programu oraz do celów testowych

    jmp wait_loop            
         ;skocz do pętli oczekującej

receive_l:

       bclr #gdd1,x:gpior           

;sygnalizacja działania programu oraz do celów testowych

          move x:rx0,b              

;pobierz próbkę sygnału wejściowego

                clr a x0,x:(r0)+ y:(r4)+,y0              ;pobranie x(n), wyczyszczenie a i pobranie wspolczynnika

                .loop #L-1



                        mac x0,y0,a x:(r0)+,x0 y:(r4)+,y0       ; obliczenie x_est

                .endl

                macr x0,y0,a


                sub a,b                                             ;wyznaczenie e(n)

                move    b,x:tx0                                ;wyślij obliczoną próbkę na wyjście (e(n))

                move #alfa,x1 b,y1

 ; oblicz 


                mpy x1,y1,a


 ;alfa * e(n)

                move a,y0

                .loop #L

                        move x:(r0)+,x0 y:(r4),a                ; wyznaczenie nowego wektora f

                        mac x0,y0,a

                        move a,y:(r4)+




                .endl

                move (r0)-

    bSET #gdd1,x:gpior        ; sygnalizacja działania programu oraz do celów testowych

 jmp wait_loop              ;jump to idle loop

